第八章　方程求根

计算机程序解方程，只能给出近似解，不能给出解析解。一般分两步进行，第一步求根的初值或存在范围，第二步求精确解。

§8.1 根的初值和存在范围

一般可靠虑以下几种方法。

8.1.1 根据方程的数学性质进行判断。

比如方程：

$$F(x)+\frac{\sqrt{x-a}+b}{\sqrt{b-x}}=0$$

二次根下的量大于等于0，分母不为0，

x-a≥0　b-x＞0　a≤x＜b.

8.1.2根据方程的物理意义进行估计。

可根据方程的物理意义来估计根的值或范围，比如，若方程中变量x代表系统中某种化学物质的摩尔分数，那么必有$0 \leqslant x \leqslant 1$。

又如实际气体状态方程

$F（P，V，T）=0$ 是一个复杂的PVT关系（化工热力学？），指定两个量，可用理想气体状态方程

$$PV=nRT$$

求第三个量的初值。

8.1.3图解法。

对f（x）=0，直角坐标系中做f（x）→x曲线, 与x轴交点就是满足方程的根. 如图:





也可把f(x)=0改成$f_1(x)=f_2(x)$,

$f_1(x)$→x与$f_2(x)$→x分别作图，两曲线交点就是方程的根. 图

8.1.4. 迈步法.



函数f（x）在a，b间连续，且f（a）与f（b）符号不同，

x=a，a+h，a+2h……时的函数值

f（a），f（a+h），f（a+2h）……,

直至两个相邻函数不具有相同的符号:

$$f(x) * f(x+h) \leq 0$$

注意h不能选择太大。

如图，若函数f（x）在a，b间连续，且f（a）与f（b）符号不同，则曲线在x=a和x=b之间至少要经过一次横坐标轴，即在a与b间至少有一个实根。

我们可以选定一个量h，称做步长，然后计算x=a，a+h，a+2h……时的函数值f（a），f（a+h），f（a+2h）……，



直至两个相邻函数不具有相同的符号。于是所求的根必定在相应的两个x之间。这种方法实际上就是从左端点x=a出发，按步长h迈步。每迈一步，检查起点x和终点x+h的函数值是否同号，如不同号，即f（x）*f（x+h）≤0，则所求根必在此x～x+h之间。这种方法称为迈步法，程序在第二节介绍。

应用此法时应注意h不能选择太大。以免在x与x+h间含有两个或两个以上的根。若h过大，在x与x+h间有偶数个根，则f（x）与f（x+h）不异号，这些根会被漏掉。

§8.2　二分法

求方程精确解的一种方法

8.2.1 方法概述

用迈步法或其他方法求出根的大致位置或存在区间后，要进一步求其精确解，方法很多，二分法是常用的方法之一。

二分法过程大致如下：设已知方程f（x）=0在$x=x_1$和x= x_2之间有且只有一个根，现取x_1与 x_2的中点x:

$$x=(x_1+x_2)/2$$

将区间分为相等的两部分，然后检查f（x_1）与f（x）的符号是否相同。





如图：如不同，则根必位于x_1与x之间，于是改为将x赋给变量x_2 ： x_2 =x ； 如相同，则根必在x和x_2之间，于是将x赋给变量x_1： x_1= x 再取新的x_1与x_2的中点，重复以上步骤，直至f（x）的绝对值小于指定值ε_1（称为函数值容差）或x_1与x_2间距离小于指定值ε_2（称为根的容差）为止。

8.2.2、 二分法框图和程序



X>B+E2
结束

方程f(x)=0
的具体形式



图 8-4 二分法程序框图

迈步法求根的初值

f(x)*f(x+h)>0?
是,继续迈步;否,
x1=x-h, x2=x…
迈步x>B+H结束

二分法求精确解

```
8010 N = 0
8020 X = A
8030 GOSUB 20000
8035 IF ABS(Y)<E1 THEN 8280
8040 X =X+H
8050 Y1=Y
8060 IF X>B+H THEN RETURN
8070 GOSUB 20000
8075 IF ABS(Y)<E1 THEN 8280
8080 IF Y1*Y>0 THEN GOTO 8040
8200 X1=X-H: X2=X
8220 X=(X1+X2)/2: GOSUB 20000
8225 IF ABS(Y)<E1 THEN GOTO 8280
8230 IF ABS(X1- X2)<E2 THEN GOTO 8280
8240 IF Y1* Y<0 THEN X2= X: GOTO 8220
8250 X1=X: Y1=Y: GOTO 8220
8280 IF X>B+E2 THEN RETURN
8290 N=N+1: XX(N)=X: A=X+H/2: GOTO 8020
```

1. 主程序中应输入的量：A：求解区间左端值a，
B：求解区间右端值b，
H：迈步步长h，
E1：函数值容差ε_1
E2：根的容差ε_2

2. 子程序输出量

N：根的个数，
XX（1），...XX（N） →N个根的值.

3. 迈步法求根的初值.

4. 二分法求精确解.

20000 Y=F(X)

20010 RETURN

上面的程序包含了迈步法求初值和二分法求精确解两步。对于一些有实际意义的数理方程，比如气体的PVT关系等，在**a、b之间有且只有一个根**，我们可以把迈步法求初值这一步去掉，只用二分法求精确解。子程序如下：

主程序中输入：**A**，**B**，**E1**，**E2**；子程序输出：**XX（1）**

```
8010 X1 = A
8020 X=A
8030 GOSUB 20000
8040 IF ABS(Y)<E1 THEN 8290
8050 Y1 = Y: X2 = B
8220 X=(X1+X2)/2:GOSUB 20000
8230 IF ABS(X1-X2)<E2 THEN GOTO 8290
8235 IF ABS(Y)< E1 THEN 8290
8240 IF Y1*Y<0 THEN X2=X: GOTO 8220
8250 X1=X: Y1=Y: GOTO 8220
8290 XX(1) = X: RETURN
```




```
20000 Y=F(X)
20010 RETURN
```

8.2.3. 应用示例　实际气体状态方程.

116页例8.3 请用范德华方程

$$(p+a/V^2)(V-b)=RT$$

计算T=313.2K，$p=5.066\times10^6$Pa，1mol二氧化碳气体所占体积（L）。对于二氧化碳，式中的常数为：$a=3.65\times10^5 L^2\cdot Pa\cdot mol^{-2}$，$b=0.0428 L\cdot mol^{-1}$，$R=8315 L\cdot Pa\cdot mol^{-1}\cdot K^{-1}$.

解：方程改写为：

$$V^3-(b+RT/p)V^2+Va/p-ab/p=0=f(V)=y$$

左端点值A：A=0。气体体积大于0。右端点值B：B=1。因按理想气体算$V=nRT/P\approx0.5$。

```
5 DIM XX(1)
10 READ A, B, H, E1, E2
20 READ T, P, AA, BB, R
30 GOSUB 8010
40 PRINT "V=";INT(XX(1)*10000!+.5)/10000!
50 END
60 DATA 0,1,0.1,1E-5,1E-5
70 DATA 313.2,5.066E6,3.65E5,0.0428,8315
8010
……
8290
20000 Y=-AA*BB/P+X*(AA/P+X*(-BB-R*T/P+X))
20010 RETURN
```

1E4

```
40 PRINT "V="; XX(1)
```

运行结果：

V=.3937988

小数点后第五位四舍五入

运行结果：

V=.3938

$$V^3-(b+RT/p)V^2+Va/p-ab/p=0=f(V)=y$$

§8.3　迭代法

8.3.4.程序

迭代法解方程, $f(x)=0$ ， 改为

$x=\varphi(x)$　x_0(给定)　x_1　$x_2\dots$

$|x_{i+1}-x_i|<EP(E2)$为止.

程序比较简单.



主程序中输入: X0, EP, N

子程序输出:收敛, X=…;

没收敛:I>N, X=…x没有达到容差要求.

```
8610 I = 1
8620 GOSUB 20000
8630 IF ABS(X - X0) < EP THEN PRINT "X="; X: GOTO 8660
8640 IF I > N THEN PRINT "I>N": PRINT "X="; X: GOTO 8660
8650 X0 = X: I = I + 1: GOTO 8620
8660 RETURN
20000 X = φ(X0): RETURN
```

8.3.5.应用示例:弱酸溶液的pH值.

例8.4 已知一价弱酸HA的电离常数为K，试求浓度为c的该弱酸溶液的pH值。$K=1.75\times10^{-5}$，$c=0.01mol/L$。

解：反应：$HA=H^{+}+A^{-}$

达到平衡时，若氢离子浓度$[H^{+}]=x$，则$[A^{-}]=x$，$[HA]=c-x$。
$[H^{+}][A^{-}]/[HA]=x^{2}/(c-x)=K$

$$x^{2}+Kx-Kc=0。$$

若公式法解：$x=[-K\pm(K^{2}+4Kc)^{1/2}]/2$；我们用迭代法，把式改写为：$x=(Kc-Kx)^{1/2}=[K(c-x)]^{1/2}$

选初值：对于弱酸，若氢离子浓度$[H^{+}]=x$较小，即$c-x\approx c$，所以：$x_{0}=(Kc)^{1/2}$。 选**EP=1E-12, N=20.** 程序:

```
10 READ K, C, EP, N
20 DATA .0000175, .01, 1E-12, 20
30 X0 = SQR(K * C)
40 GOSUB 8610
50 D = SQR(K * (K + 4 * C))
60 X1 = .5 * (-K + D)
70 X2 = .5 * (-K - D)
80 PRINT "X1="; X1
90 PRINT "X2="; X2
100 PRINT
110 PRINT "PH="; -LOG(X) / LOG(10)
120 END
8610 --8660
20000 X = SQR(K * (C - X0))
20010 PRINT X0: RETURN
```

运行结果:

X=4.096715E-04

X1= 4.096715E-04

X2=-4.271715E-04

pH=3.387564

输出若I＞N, X没有达到容差要求， 怎么办?

1、增大迭代次数，如把N=20改为N=40。

2、X= φ(x)可有多种形式,一种不收敛,可换另一种试.
如: $x^2+Kx-Kc=0$, 可以写成：$x=[K(c-x)]^{1/2}$;
也可以写成: $x=(Kc-x^2)/K$

作业: 131页 一.

.

习题：

一：请编程用二分法计算 1 mol/L $[Ni(NH_3)_6](NO_3)_2$溶液中Ni^{2+}的浓度，已知络合反应

$$Ni(NH_3)_6^{2+} \risingdotseq Ni^{2+}+6NH_3$$

的不稳定常数$K_d=5.7X10^{-9}$。