



第十章 插值

我们在化学、化工文献手册中查到的,以及在实验中测定的数据,一般总是一组离散数据。比如一组 (x_i, y_i) , 绘在坐标图上, 得一组离散点, 这些点称为 **节点**。设有若干个节点 (x_i, y_i) , 现根据这些节点绘出一条光滑的曲线, 也就是构造逼近函数

$$y=f(x)$$

以近似表示 x 与 y 之间的关系, 如此, 各不同 x 时的 y 值, 就可根据逼近函数 $f(x)$ 求出。





如果绘制曲线或构造函数时，要求曲线严格通过各节点,即要求 $y_i=f(x_i)$ ，这就是插值问题；如果只要求曲线在总体上与各节点相符合而不要求曲线严格通过各节点，这就是拟合问题。

本章讨论插值问题。

§ 10.1 拉格朗日一元全节点插值

10.1.1、一般介绍：

一元是指只有一个自变量 x 和一个应变量 y ，
全节点是指利用全部节点。

一般来说，若与一条曲线相应的方程内含有 M 个可调参数，则该曲线必能严格通过 M 个节点，
从而可用以进行插值。

比如有一曲线，其相应的方程为

$$y=b_0+b_1x+b_2x^2+\dots+b_nx^n \quad (10.1)$$

式中 $b_0 \sim b_n$ 为 $n+1$ 个可调参数. 该曲线必严格通过
 $n+1$ 个节点 $(x_0, y_0) \sim (x_n, y_n)$ 。

这是因为，若将这 $n+1$ 个节点坐标值代入10.1式, 可得联立方程组10.2:

$$\begin{aligned} y_0 &= b_0 + b_1 x_0 + b_2 x_0^2 + \dots + b_n x_0^n \\ y_1 &= b_0 + b_1 x_1 + b_2 x_1^2 + \dots + b_n x_1^n \\ &\dots\dots\dots \\ y_n &= b_0 + b_1 x_n + b_2 x_n^2 + \dots + b_n x_n^n \end{aligned} \quad (10.2)$$

$n+1$ 个方程, $n+1$ 个未知数, 可解出每一个参数。如此求出的 b_0 , $b_1 \dots b_n$ 作为方程组的解, 必能满足方程组内的任一个方程, 这就意味着, 与这组参数相应的曲线必严格通过这确定的 $n+1$ 个节点。所以, 通过 $n+1$ 个点的插值函数, 应含有 $n+1$ 个可调参数。但若直接用式10.1右方函数做插值函数, 必须求解多元联立方程组来确定函数中的参数, 很麻烦。

我们不用10.1式函数，而用拉格朗日插值函数。它含有 $n+1$ 个参数，插值时无须求出这些参数。

10.1.2、一元全节点插值的算法

拉格朗日插值函数的形式为：

$$\begin{aligned} y = & a_0(x-x_1)(x-x_2)\dots(x-x_n) \\ & + a_1(x-x_0)(x-x_2)\dots(x-x_n) \\ & + \dots\dots\dots \\ & + a_n(x-x_0)(x-x_1)\dots(x-x_{n-1}) \end{aligned}$$

写成连加连乘的形式为：

$$y = \sum_{i=0}^n \left[a_i \prod_{\substack{j=0 \\ (j \neq i)}}^n (x - x_j) \right] \quad (10.3)$$

10.3式含有 $n+1$ 个可调参数 $a_0, a_1 \dots a_n$ ，其相应的曲线必通过 $n+1$ 个节点。下面通过运算，把 a_i 用节点坐标表示出来。

将 (x_0, y_0) 代入： $y_0 = a_0(x_0 - x_1)(x_0 - x_2) \dots (x_0 - x_n)$

(x_1, y_1) 代入： $y_1 = a_1(x_1 - x_0)(x_1 - x_2) \dots (x_1 - x_n)$

将 (x_i, y_i) 代入得:

$$y_i = a_i (x_i - x_0) (x_i - x_1) \dots (x_i - x_{i-1}) (x_i - x_{i+1}) \dots (x_i - x_n)$$

$$y_i = a_i \prod_{\substack{j=0 \\ (j \neq i)}}^n (x_i - x_j)$$

$$a_i = y_i / \left[\prod_{\substack{j=0 \\ (j \neq i)}}^n (x_i - x_j) \right] \quad (10.4)$$

这样，我们求出了待定参数与坐标点间的关系。将10.4代回10.3得：

$$y = \sum_{i=0}^n \left[y_i \prod_{\substack{j=0 \\ (j \neq i)}}^n (x - x_j) \right] / \left[\prod_{\substack{j=0 \\ (j \neq i)}}^n (x_i - x_j) \right] = \sum_{i=0}^n \left[y_i \prod_{\substack{j=0 \\ (j \neq i)}}^n \frac{x - x_j}{x_i - x_j} \right] \quad (10.5)$$

这样，我们求出了待定参数与坐标点间的关系。

（10.5）就是拉格郎日一元全节点插值公式，该式右方只含节点坐标和被插值点自变量 x ，可用它直接求 y 。

拉格郎日插值法体现的是两个变量间的关系。任何一个变量都可取做自变量。如果我们取y做自变量，则按照与上述相同步骤可得：

$$x = \sum_{i=0}^n \left[x_i \prod_{\substack{j=0 \\ (j \neq i)}}^n \frac{y - y_j}{y_i - y_j} \right] \quad (10.6)$$

求出各指定y值时的x值，这样的插值称为反插。

10.1.3、一元全节点插值的程序

以子程序形式给出。

输入量：N：节点编号终点n(编号从0开始)。

X(0) ..., X(N) 节点自变量 $X_0 \dots X_n$.

Y(0) ..., Y(N) 节点应变量是 $Y_0 \dots Y_n$.

XX---被插值点自变量值X .

输出量是YY---被插值点应变量值Y.

10010 YY = 0

10020 FOR I = 0 TO N

10030 L = 1

10040 FOR J = 0 TO N

10050 IF J<>I THEN L=L*(XX-X(J))/(X(I)-X(J))

10060 NEXT J

10070 YY = YY + L * Y(I)

10080 NEXT I

10090 RETURN

$$y = \sum_{i=0}^n \left[y_i \prod_{\substack{j=0 \\ (j \neq i)}}^n \frac{x - x_j}{x_i - x_j} \right] \quad (10.5)$$

10.1.4. 应用示例.

热容---温度数据的插值之一： 已知乙炔在不同温度($t/^{\circ}\text{C}$)下的等压热容数据($C_p/\text{J}\cdot\text{mol}\cdot\text{K}^{-1}$)如表10-1所示。请用拉格朗日全节点插值公式求 350°C 的热容数据。

i	t	C_p	i	t	C_p
0	0	42.92	5	400	60.25
1	25	44.80	6	500	62.84
2	100	49.45	7	700	67.16
3	200	53.93	8	900	70.76
4	300	57.49	9	1127	73.81

输入量：N：节点编号终点 n (编号从0开始)。N=9。

$X(0) \dots, X(N)$ 节点自变量 $X_0 \dots X_n$ 。

$Y(0) \dots, Y(N)$ 节点应变变量是 $Y_0 \dots Y_n$ 。

XX---被插值点自变量值 X_0 。XX=350。 程序：

运行结果:

CP(T= 350)=58.94932

10 READ N, XX

20 DIM X(N), Y(N)

30 FOR I = 0 TO N: READ X(I), Y(I): NEXT I

40 GOSUB 10010

50 PRINT "CP(T="; XX; ")="; YY

60 END

200 DATA 9,350, 0,42.92,25,44.8,100,49.45

210 DATA 200,53.93,300,57.49, 400,60.25

220 DATA 500,62.84,700,67.16,900,70.76,1127,73.81

10010---10090

§ 10.2 一元部分节点插值

在实际工作中,插值可不必使用全部节点,只使用靠近被插值点的部分节点即可。

这是因为: 1、远离插值点的节点对插值的影响较小;
2、节省计算量。

10010 $YY = 0$

10020 FOR I = 0 TO N

10030 L = 1 FOR II = N1 TO N2

FOR JJ = N1 TO N2 II, JJ, LL

10040 FOR J = 0 TO N

10050 IF $J \neq I$ THEN $L = L * (XX - X(J)) / (X(I) - X(J))$

10060 NEXT J

10070 $YY = YY + L * Y(I)$

10080 NEXT I

10090 RETURN

为了利用部分节点进行插值,最简单的办法,就是不将节点编号限为0至N,而是定为N1至N2. 为此,只需将全节点程序做如左方修改:

这相当于把(10.5)式

$$y = \sum_{i=0}^n \left[y_i \prod_{\substack{j=0 \\ (j \neq i)}}^n \frac{x - x_j}{x_i - x_j} \right]$$

修改为:

$$y = \sum_{i=n_1}^{n_2} \left[y_i \prod_{\substack{j=n_1 \\ (j \neq i)}}^{n_2} \frac{x - x_j}{x_i - x_j} \right] \quad (10.5')$$

作为一元部分节点插值的公式。

修改后的程序：

```
10310 YY = 0
10320 FOR II = N1 TO N2
10330 LL = 1
10340 FOR JJ = N1 TO N2
10350 IF JJ<>II THEN LL=LL*
      (XX-X(JJ))/(X(II)-X(JJ))
10360 NEXT JJ
10370 YY = YY + LL * Y(II)
10380 NEXT II
10390 RETURN
```

本 子 程 序 输 入 量
N1, N2, X(0)–X(N)
Y(0)–Y(N) XX
插值点自变量.

输出量YY 被插值
点Y值.

N1—N2应是最靠
近插值点的几个数.
需要自己判断.

10.2.1 应用示例.

热容--温度数据的插值之一：已知乙炔在不同温度($t/^{\circ}\text{C}$)下的等压热容数据($C_p/\text{J}\cdot\text{mol}\cdot\text{K}^{-1}$)如表10-1所示. 请用拉格朗日全节点插值公式利用部分节点求 350°C 的热容数据.

i	t	C_p	i	t	C_p
0	0	42.92	5	400	60.25
1	25	44.80	6	500	62.84
2	100	49.45	7	700	67.16
3	200	53.93	8	900	70.76
4	300	57.49	9	1127	73.81

输入量：N：节点编号终点n(编号从0开始)。N=9. $X(0) \dots, X(N)$ 节点自变量 $X_0 \dots X_n$ 。 $Y(0) \dots, Y(N)$ 节点应变变量是 $Y_0 \dots Y_n$ 。 XX--被插值点自变量值 X_0 。 XX=350。 若用最靠近插值点的4个温度进行插值，XX=350上推两个下推两个， 200°C — 500°C ，N1=3，N2=6， 程序：


```

10 READ N, N1, N2, XX
20 DIM X(N), Y(N)
30 FOR I=0 TO N:READ X(I),Y(I):NEXT I
40 GOSUB 10310
50 PRINT "CP(T=";XX;")=";YY
60 END      3,6
200 DATA 9,0,9,350,0,42.92,25,44.8
210 DATA 100,49.45,200,53.93,300,57.49
220 DATA 400,60.25,500,62.84,
      700,67.16,900,70.76,1127,73.81
10310 YY = 0
.....
10390 RETURN

```

运行结果: 0,9:
 CP(T= 350)=58.94932
 3,6
 CP(T= 350)=58.93062
 2,7
 CP(T= 350)=58.932

作业：习题一。拉格朗日一元全节点插值。

习题：一、已知水的密度 ρ 与温度 t 见的的数据如下：

$t/^{\circ}\text{C}$	40	50	60	70	80
$\rho/\text{g}\bullet\text{ml}^{-1}$	0.9922	0.9881	0.9832	0.9778	0.9718
	90	0.9653			

请用拉格朗日插值法求 $t=65^{\circ}\text{C}$ 时的 ρ 。

§ 10.3. 拉格朗日二元插值


设有如下数表, X_i 与 Y_j 为插值节点的自变量, Z_{ij} 为插值节点的应变变量, 现在的问题是, 被插值点自变量分别为 X 和 Y 时, 应变变量 Z 值.

我们直接给出拉格朗日二元插值计算公式:

$y \backslash x$	5	10	15	20
5	3.45	3.30	3.24	3.21
10	2.52	2.32	2.24	2.20
15	2.27	2.06	1.97	1.92
20	2.16	1.94	1.89	1.79

$$z = \sum_{i=n_1}^{n_2} \sum_{j=m_1}^{m_2} \left[z_{ij} \prod_{\substack{k=n_1 \\ (k \neq i)}}^{n_2} \frac{x - x_k}{x_i - x_k} \prod_{\substack{l=m_1 \\ (l \neq j)}}^{m_2} \frac{y - y_l}{y_j - y_l} \right] \quad (10.11)$$

和一元部分节点插值的公式（10.5'）类似。
n1、n2为自变量**x**起点和终点编号； **m1、m2**
 为自变量**y**起点和终点编号。



$$z = \sum_{i=n_1}^{n_2} \sum_{j=m_1}^{m_2} \left[z_{ij} \prod_{\substack{k=n_1 \\ (k \neq i)}}^{n_2} \frac{x - x_k}{x_i - x_k} \prod_{\substack{l=m_1 \\ (l \neq j)}}^{m_2} \frac{y - y_l}{y_j - y_l} \right]$$

10500 ZZ = 0

10510 FOR I = N1 TO N2

10520 FOR J = M1 TO M2

10530 A = 1

10540 FOR K = N1 TO N2

10550 IF K<>I THEN A=A*(XX-X(K))/(X(I)-X(K))

10560 NEXT K

10570 FOR L = M1 TO M2

10580 IF L<>J THEN A=A*(YY-Y(L))/(Y(J)-Y(L))

10590 NEXT L

10600 ZZ = ZZ + A * Z(I, J)

10610 NEXT J

10620 NEXT I

10630 RETURN

输入量: N1, N2, M1, M2,

X(N1)-X(N2), Y(M1)-Y(M2)

Z(N1,M1)-Z(N2,M2), XX, YY

输出量: ZZ

10.3.3 应用示例:

数表插值. 164页. 计算 $x=7, y=11$ 时的函数值

$F(7,11)=?$

输入量: x, y 值都是4个, 编号从1开始, 取 $N1=1, N2=4, M1=1, M2=4, X(N1)-X(N2), Y(M1)-Y(M2), Z(N1, M1)-Z(N2, M2)$ 都在表中. $XX=7, YY=11$

$y \backslash x$	5	10	15	20
5	3.45	3.30	3.24	3.21
10	2.52	2.32	2.24	2.20
15	2.27	2.06	1.97	1.92
20	2.16	1.94	1.89	1.79

程序:

```
10 READ N1, N2, M1, M2, XX, YY
20 DIM X(N2), Y(M2), Z(N2, M2)
30 FOR I=N1 TO N2:READ X(I):NEXT I
40 FOR J=M1 TO M2:READ Y(J):NEXT J
50 FOR I=N1 TO N2:FOR J=M1 TO M2:READ Z(I,J)
   :NEXT J:NEXT I
60 GOSUB 10500
70 PRINT "F("; XX; ", "; YY; ")="; ZZ
80 END   200 DATA 1,4,1,4,7,11
210 DATA 5,10,15,20   220 DATA 5,10,15,20
240 DATA 3.45,2.52,2.27,2.16  250 DATA 3.30,2.32,2.06,1.94
260 DATA 3.24,2.24,1.97,1.89  270 DATA 3.21,2.2,1.92,1.79
10500 ..... 10630
```

运行结果:

F(7,11)=2.332036

上面是书上的程序， x 、 y 编号从1开始，而一元节点插值编号从0开始。对应于一元部分节点插值的程序， x 、 y 编号若从0开始，拉格朗日二元插值程序修改如下：

```
10 READ N, N1, N2, M, M1, M2, XX, YY
20 DIM X(N), Y(M), Z(N, M)
30 FOR I=0 TO N:READ X(I):NEXT I
40 FOR J=0 TO M:READ Y(J):NEXT J
50 FOR I=0 TO N:FOR J=0 TO M:READ Z(I,J):NEXT J:NEXT I
60 GOSUB 10500
70 PRINT "F("; XX; ", "; YY; ")="; ZZ
80 END  200 DATA 3,0,3,3,0,3,7,11  210.....
```

运行结果:

200 DATA 3,0,3,3,0,3,7,11

F(7,11)=2.332036

200 DATA 3,0,2,3,0,3,7,11

F(7,11)=2.336931

200 DATA 3,0,3,3,0,2,7,11

F(7,11)=2.31361

.....